A DEVICE-INDEPENDENT MACRO TO FORCE THE SAS/GRAPH® PLOTTING AREA TO BE SQUARE

Earl R. Westerlund
Eastman Kodak Company

In many kinds of statistical plots, it is important that the plot produced be exactly square. Unfortunately, SAS/GRAPH® software does not provide a simple option to ensure that the plotting area is square for any device. If a site has many different kinds of devices, it can be difficult for users to know the proper HSIZE and VSIZE GOPTIONS to set.

One way to handle this problem is to take advantage of the DATA Step Graphics Interface (DSGI). DSGI is an interface to the graphics routines used by SAS/GRAPH software to create a custom graph or enhance an existing graph. One of the interface routines available is the GASK routine, which allows one to check the current settings of a variety of values, including the current HSIZE and VSIZE. By using the GASK routine to check the HSIZE and VSIZE values for the current device, we can set the maximum square viewing area.

A macro to do this, and illustrative examples, follow.

```
%macro setsq;
   /**** Reset HSIZE and VSIZE to defaults for this device *****/
   GOPTIONS HSIZE= VSIZE=;

   DATA _NULL_;  
      /**** Invoke DSGI *****/
      RC = GINIT('');  
      RC = GRAPH('CLEAR', '_DUMMY');  
      * Initialize;  
      * Create dummy graph;

      /**** Get HSIZE and VSIZE *****/
      CALL GASK('HSIZE', HSIZE, RC);  
      CALL GASK('VSIZE', VSIZE, RC);  
      PSIZE = MIN(HSIZE, VSIZE);  
      CALL SYMPUT('PSIZE', PUT(PSIZE,BEST12.));  
      * Save min size;

      /**** Update DSGI, but do not display anything *****/
      RC = GRAPH('UPDATE', 'NOSHOW');  
      RC = GRAPH('DELETE', '_DUMMY');  
      RC = GTERM();  
      * Display non-graph;  
      * Delete non-graph;  
      * Exit DSGI;

      RUN;

      /**** Use minimum size to fix a square plotting area *****/
      GOPTIONS HSIZE=&psize VSIZE=&psize;

%mend setsq;
```
Examples

Replayed Plot without Using %SETSQ

Replayed Plot Using %SETSQ

Reference

SAS/GRAPH is a registered trademark of SAS Institute Inc. No endorsement of this product by Eastman Kodak Company is implied. The opinions expressed in this paper are those of the author, and are not representative of Eastman Kodak Company.

Author
The author may be contacted at Internet address earlw@kodak.com.

NFSUG '93 Proceedings